

---

# ■ Solidity Project ■

## Shared Wallet with Allowance Function

Copyright © 2019-2020 Thomas Wiesner – <https://vomtom.at>

The included Source-Code shall be Distributed under the License: "[Attribution CC BY 4.0](#)"

---

### Real-World Use-Case for this Project

- Allowance for Children per day/week/month to be able to spend a certain amount of funds.
- Employers give employees an allowance for their travel expenses.
- Businesses give contractors an allowance to spend a certain budget.

### Development-Goal

- Have an on-chain wallet smart contract.
- This wallet contract can store funds and let users withdraw again.
- You can also give "allowance" to other, specific user-addresses.
- Restrict the functions to specific user-roles (owner, user)
- Re-Use existing smart contracts which are already audited to the greatest extent

### Contents

Step 1 – We Define the Basic Smart Contract.....	2
Step 2 – Permissions: Allow only the Owner to Withdraw Ether .....	2
Step 3 – Permissions: Use Re-Usable Smart Contracts from OpenZeppelin .....	3
Step 4 – Permissions: Add Allowances for External Roles .....	3
Step 5 – Improve/Fix Allowance to avoid Double-Spending .....	4
Step 6 – Improve Smart Contract Structure.....	5
Step 7 – Add Events in the Allowances Smart Contract.....	6
Step 8 – Add Events in the SharedWallet Smart Contract .....	7
Step 9 – Add the SafeMath Library safeguard Mathematical Operations.....	8
Step 10 – Remove the Renounce Ownership functionality .....	8
Step 11 – Move the Smart Contracts into separate Files .....	9
The Final Smart Contract(s) .....	10
File: Allowance.sol.....	10
File: SmartContract.sol.....	11

## Step 1 – We Define the Basic Smart Contract

This is the very basic smart contract. It can receive Ether and it's possible to withdraw Ether, but all in all, not very useful quite yet. Let's see if we can improve this a bit in the next step. Open the Remix IDE and follow along:

```
pragma solidity ^0.6.0;

contract SharedWallet {

    function withdrawMoney(address payable _to, uint _amount) public {
        _to.transfer(_amount);
    }

    receive () external payable {

    }

}
```

## Step 2 – Permissions: Allow only the Owner to Withdraw Ether

In this step we restrict withdrawal to the owner of the wallet. How can we determine the owner? It's the user who deployed the smart contract.

```
pragma solidity ^0.6.0;

contract SharedWallet {

    address owner;

    constructor() internal {
        owner = msg.sender;
    }

    modifier onlyOwner() {
        require(msg.sender == owner, "You are not allowed");
        _;
    }

    function withdrawMoney(address payable _to, uint _amount) public onlyOwner {
        _to.transfer(_amount);
    }

    Receive () external payable {

    }

}
```

## Step 3 – Permissions: Use Re-Usable Smart Contracts from OpenZeppelin

Having the owner-logic directly in one smart contract isn't very easy to audit. Let's break it down into smaller parts and re-use existing audited smart contracts from OpenZeppelin for that.

```
pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";

contract SharedWallet is Ownable {

    function withdrawMoney(address payable _to, uint _amount) public onlyOwner {
        _to.transfer(_amount);
    }

    receive() external payable {

    }
}
```

**ATTENTION: Solidity Update 0.6 and OpenZeppelin Update 0.3: The Ownable smart contract moved to [/access/Ownable.sol](https://github.com/OpenZeppelin/openzeppelin-contracts/contracts/access/Ownable.sol).**

## Step 4 – Permissions: Add Allowances for External Roles

In this step we are adding a mapping so we can store address => uint amounts. This will be like an array that stores [0x123546...] an address, to a specific number. So, we always know how much someone can withdraw. We also add a new modifier that checks: Is it the owner itself or just someone with allowance?

```

pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";

contract SharedWallet is Ownable {
    mapping(address => uint) public allowance;

    function addAllowance(address _who, uint _amount) public onlyOwner {
        allowance[_who] = _amount;
    }

    modifier ownerOrAllowed(uint _amount) {
        require(msg.sender == owner() || allowance[msg.sender] >= _amount, "You are not allowed!");
        _;
    }

    function withdrawMoney(address payable _to, uint _amount) public ownerOrAllowed(_amount) {
        _to.transfer(_amount);
    }

    receive() external payable {
    }
}

```

**ATTENTION OpenZeppelin Update 0.3: the function isOwner() was removed. I have substituted it with msg.sender == owner()**

## Step 5 – Improve/Fix Allowance to avoid Double-Spending

Without reducing the allowance on withdrawal, someone can continuously withdraw the same amount over and over again. We have to reduce the allowance for everyone other than the owner.

```

function reduceAllowance(address _who, uint _amount) internal ownerOrAllowed(_amount) {
    allowance[_who] -= _amount;
}

function withdrawMoney(address payable _to, uint _amount) public ownerOrAllowed(_amount) {
    require(_amount <= address(this).balance, "Contract doesn't own enough money");
    if(msg.sender != owner()) {
        reduceAllowance(msg.sender, _amount);
    }
    _to.transfer(_amount);
}

```

## Step 6 – Improve Smart Contract Structure

Now we know our basic functionality, we can structure the smart contract differently. To make it easier to read, we can break the functionality down into two distinct smart contracts.

```
pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";

contract Allowance is Ownable {

    mapping(address => uint) public allowance;

    function setAllowance(address _who, uint _amount) public onlyOwner {
        allowance[_who] = _amount;
    }

    modifier ownerOrAllowed(uint _amount) {
        require(msg.sender == owner() || allowance[msg.sender] >= _amount, "You are not al
lowed!");
        _;
    }

    function reduceAllowance(address _who, uint _amount) internal ownerOrAllowed(_amount)
{
        allowance[_who] -= _amount;
    }
}

contract SharedWallet is Ownable, Allowance {

    function withdrawMoney(address payable _to, uint _amount) public ownerOrAllowed(_amoun
t) {
        require(_amount <= address(this).balance, "Contract doesn't own enough money");
        if(msg.sender != owner()) {
            reduceAllowance(msg.sender, _amount);
        }
        _to.transfer(_amount);
    }

    receive() external payable {

    }
}
```

## Step 7 – Add Events in the Allowances Smart Contract

One thing that's missing is events.

```
pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";

contract Allowance is Ownable {

    event AllowanceChanged(address indexed _forWho, address indexed _byWhom, uint _oldAmount, uint _newAmount);
    mapping(address => uint) public allowance;

    function setAllowance(address _who, uint _amount) public onlyOwner {
        emit AllowanceChanged(_who, msg.sender, allowance[_who], _amount);
        allowance[_who] = _amount;
    }

    modifier ownerOrAllowed(uint _amount) {
        require(msg.sender == owner() || allowance[msg.sender] >= _amount,
            "You are not allowed!");
        _;
    }

    function reduceAllowance(address _who, uint _amount) internal ownerOrAllowed(_amount)
    {
        emit AllowanceChanged(_who, msg.sender, allowance[_who], allowance[_who] - _amount);
        allowance[_who] -= _amount;
    }
}

contract SharedWallet is Ownable, Allowance {
    //...
}
```

## Step 8 – Add Events in the SharedWallet Smart Contract

```
contract SharedWallet is Ownable, Allowance {

    event MoneySent(address indexed _beneficiary, uint _amount);
    event MoneyReceived(address indexed _from, uint _amount);

    function withdrawMoney(address payable _to, uint _amount) public ownerOrAllowed(_amount) {
        require(_amount <= address(this).balance, "Contract doesn't own enough money");
        if(msg.sender != owner()) {
            reduceAllowance(msg.sender, _amount);
        }
        emit MoneySent(_to, _amount);
        _to.transfer(_amount);
    }

    receive() external payable {
        emit MoneyReceived(msg.sender, msg.value);
    }
}
```

## Step 9 – Add the SafeMath Library safeguard Mathematical Operations

```
pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";
import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/math/SafeMath.sol";

contract Allowance is Ownable {

    using SafeMath for uint;

    event AllowanceChanged(address indexed _forWho, address indexed _byWhom, uint _oldAmount, uint _newAmount);
    mapping(address => uint) public allowance;

    function setAllowance(address _who, uint _amount) public onlyOwner {
        //...
    }

    modifier ownerOrAllowed(uint _amount) {
        //...
    }

    function reduceAllowance(address _who, uint _amount) internal ownerOrAllowed(_amount)
    {
        emit AllowanceChanged(_who, msg.sender, allowance[_who], allowance[_who].sub(_amount));
        allowance[_who] = allowance[_who].sub(_amount);
    }
}

contract SharedWallet is Ownable, Allowance {
    //...
}
```

## Step 10 – Remove the Renounce Ownership functionality

```
contract SharedWallet is Ownable, Allowance {

    //...

    function renounceOwnership() public override onlyOwner {
        revert("can't renounceOwnership here"); //not possible with this smart contract
    }

    //...
}
```



## Step 11 – Move the Smart Contracts into separate Files

Allowance.sol:

```
pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";
import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/math/SafeMath.sol";

contract Allowance is Ownable {
    //...
}
```

SharedWallet.sol:

```
pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";
import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/math/SafeMath.sol";
import "./Allowance.sol";

contract SharedWallet is Ownable, Allowance {
    //...
}
```

## The Final Smart Contract(s)

File: Allowance.sol

```
pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";
import "https://github.com/OpenZeppelin/openzeppelin-contracts/contracts/math/SafeMath.sol";

contract Allowance is Ownable {

    using SafeMath for uint;

    event AllowanceChanged(address indexed _forWho, address indexed _byWhom, uint _oldAmount
, uint _newAmount);
    mapping(address => uint) public allowance;

    function setAllowance(address _who, uint _amount) public onlyOwner {
        emit AllowanceChanged(_who, msg.sender, allowance[_who], _amount);
        allowance[_who] = _amount;
    }

    modifier ownerOrAllowed(uint _amount) {
        require(msg.sender == owner() || allowance[msg.sender] >= _amount, "You are not allo
wed!");
        _;
    }

    function reduceAllowance(address _who, uint _amount) internal ownerOrAllowed(_amount) {
        emit AllowanceChanged(_who, msg.sender, allowance[_who], allowance[_who].sub(_amount
));
        allowance[_who] = allowance[_who].sub(_amount);
    }

    function renounceOwnership() public override onlyOwner {
        revert("can't renounceOwnership here"); //not possible with this smart contract
    }
}
```

```
pragma solidity ^0.6.0;

import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/access/Ownable.sol";
import "https://github.com/OpenZeppelin/openzeppelin-
contracts/contracts/math/SafeMath.sol";
import "./Allowance.sol";

contract SharedWallet is Ownable, Allowance {

    event MoneySent(address indexed _beneficiary, uint _amount);
    event MoneyReceived(address indexed _from, uint _amount);

    function withdrawMoney(address payable _to, uint _amount) public ownerOrAllowed(_amount) {
        require(_amount <= address(this).balance, "Contract doesn't own enough money");
        if(!isOwner()) {
            reduceAllowance(msg.sender, _amount);
        }
        emit MoneySent(_to, _amount);
        _to.transfer(_amount);
    }

    function renounceOwnership() public override onlyOwner {
        revert("can't renounceOwnership here"); //not possible with this smart contract
    }

    receive() external payable {
        emit MoneyReceived(msg.sender, msg.value);
    }
}
```

---

# Congratulations, LAB is completed



---

From the Course “Ethereum Blockchain Developer – Build Projects in Solidity”



FULL COURSE:

<https://www.udemy.com/course/blockchain-developer/?referralCode=E8611DF99D7E491DFD96>